In[1]:=
```
Needs["VectorAnalysis`"]
```

- **Extract ϵ and μ matrices from LeLmLocal notebook and rename x-> Xx since we will use package VectorAnalysis to compute curl and div.**

In[2]:=
```
LeMat = {{0, Cos[x], -λ Sin[x]}, {-Cos[x], λ Sin[x] Sin[x + ϕ], Cos[x] Sin[x + ϕ]},
    {λ Sin[x], Cos[x + ϕ] Sin[x], λ Cos[x] Cos[x + ϕ]}} /. x → Xx;
```

In[3]:=
```
LmMat = {{0, Cos[x + ϕ], -λ Sin[x + ϕ]}, {-Cos[x + ϕ], λ Sin[x] Sin[x + ϕ], Cos[x + ϕ] Sin[x]},
    {λ Sin[x + ϕ], Cos[x] Sin[x + ϕ], λ Cos[x] Cos[x + ϕ]}} /. x → Xx;
```

In[51]:=
```
LeMat // MatrixForm
```

Out[51]//MatrixForm=

$$\begin{pmatrix} 0 & \mathrm{Cos}[Xx] & -\lambda\,\mathrm{Sin}[Xx] \\ -\mathrm{Cos}[Xx] & \lambda\,\mathrm{Sin}[Xx]\,\mathrm{Sin}[Xx+\phi] & \mathrm{Cos}[Xx]\,\mathrm{Sin}[Xx+\phi] \\ \lambda\,\mathrm{Sin}[Xx] & \mathrm{Cos}[Xx+\phi]\,\mathrm{Sin}[Xx] & \lambda\,\mathrm{Cos}[Xx]\,\mathrm{Cos}[Xx+\phi] \end{pmatrix}$$

In[52]:=
```
LmMat // MatrixForm
```

Out[52]//MatrixForm=

$$\begin{pmatrix} 0 & \mathrm{Cos}[Xx+\phi] & -\lambda\,\mathrm{Sin}[Xx+\phi] \\ -\mathrm{Cos}[Xx+\phi] & \lambda\,\mathrm{Sin}[Xx]\,\mathrm{Sin}[Xx+\phi] & \mathrm{Cos}[Xx+\phi]\,\mathrm{Sin}[Xx] \\ \lambda\,\mathrm{Sin}[Xx+\phi] & \mathrm{Cos}[Xx]\,\mathrm{Sin}[Xx+\phi] & \lambda\,\mathrm{Cos}[Xx]\,\mathrm{Cos}[Xx+\phi] \end{pmatrix}$$

- **Doublecheck that LeMat coincides with expressions in paper**

In[57]:=
```
(* pricipal part of permittivity ϵ *)
Simplify[(LeMat + Transpose[LeMat]) / 2] // MatrixForm
(* skewon part of permittivity ϵ *)
Simplify[(LeMat - Transpose[LeMat]) / 2] // MatrixForm
(* pricipal part of permeability μ *)
Simplify[(LmMat + Transpose[LmMat]) / 2] // MatrixForm
(* skewon part of permeability μ *)
Simplify[(LmMat - Transpose[LmMat]) / 2] // MatrixForm
```

Out[57]//MatrixForm=

$$\begin{pmatrix} 0 & 0 & 0 \\ 0 & \lambda\,\mathrm{Sin}[Xx]\,\mathrm{Sin}[Xx+\phi] & \frac{1}{2}\mathrm{Sin}[2\,Xx+\phi] \\ 0 & \frac{1}{2}\mathrm{Sin}[2\,Xx+\phi] & \lambda\,\mathrm{Cos}[Xx]\,\mathrm{Cos}[Xx+\phi] \end{pmatrix}$$

Out[58]//MatrixForm=

$$\begin{pmatrix} 0 & \mathrm{Cos}[Xx] & -\lambda\,\mathrm{Sin}[Xx] \\ -\mathrm{Cos}[Xx] & 0 & \frac{\mathrm{Sin}[\phi]}{2} \\ \lambda\,\mathrm{Sin}[Xx] & -\frac{\mathrm{Sin}[\phi]}{2} & 0 \end{pmatrix}$$

Out[59]//MatrixForm=

$$\begin{pmatrix} 0 & 0 & 0 \\ 0 & \lambda\,\mathrm{Sin}[Xx]\,\mathrm{Sin}[Xx+\phi] & \frac{1}{2}\mathrm{Sin}[2\,Xx+\phi] \\ 0 & \frac{1}{2}\mathrm{Sin}[2\,Xx+\phi] & \lambda\,\mathrm{Cos}[Xx]\,\mathrm{Cos}[Xx+\phi] \end{pmatrix}$$

Out[60]//MatrixForm=

$$\begin{pmatrix} 0 & \mathrm{Cos}[Xx+\phi] & -\lambda\,\mathrm{Sin}[Xx+\phi] \\ -\mathrm{Cos}[Xx+\phi] & 0 & -\frac{\mathrm{Sin}[\phi]}{2} \\ \lambda\,\mathrm{Sin}[Xx+\phi] & \frac{\mathrm{Sin}[\phi]}{2} & 0 \end{pmatrix}$$

- **Note: the above computations show that both permittivity ϵ and permeability μ have a skewon component.**

- **Define fields as in Example 3.16**

In[20]:=
```
Efield = Cos[t] {0, λ Sin[ Xx], Cos[ Xx]};
Hfield = -Sin[t] {0, λ Sin[ Xx + ϕ], Cos[Xx + ϕ]};
```

In[22]:=
```
Dfield = Simplify[Simplify[LeMat.Efield] /. λ^2 → 1]
Bfield = Simplify[Simplify[LmMat.Hfield] /. λ^2 → 1]
```

Out[22]= {0, Cos[t] Sin[Xx + ϕ], λ Cos[t] Cos[Xx + ϕ]}

Out[23]= {0, -Sin[t] Sin[Xx], -λ Cos[Xx] Sin[t]}

- **Check that permittivity Le and permeability Lm are invertible**

In[41]:=
```
Simplify[Simplify[Det[LeMat] /. λ^2 → 1] /. λ^2 → 1]
Simplify[Simplify[Det[LmMat] /. λ^2 → 1] /. λ^2 → 1]
```

Out[41]= λ Cos[ϕ]

Out[42]= λ Cos[ϕ]

- **Check that Maxwell's equations hold for Efield, Hfield, Dfield and Bfield**

In[47]:=
```
Simplify[Curl[Efield] == -D[Bfield, t]]
Simplify[Curl[Hfield] == D[Dfield, t]]
Div[Dfield]
Div[Bfield]
```

Out[47]= True

Out[48]= True

Out[49]= 0

Out[50]= 0

- **Compute energy of solution**

In[62]:=
```
Simplify[Efield.Dfield + Hfield.Bfield]
```

Out[62]= λ Cos[ϕ]

- **Compute the Poynting vector of solution:**

In[63]:=
```
Simplify[Cross[Efield, Hfield]]
```

Out[63]= {λ Cos[t] Sin[t] Sin[ϕ], 0, 0}